Windows-Funktionen komfortabel aus Access aufrufen, Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

Mit VBA lassen sich zwar viele hilfreiche Zusatzfunktionen für eine Datenbankanwendung realisieren, aber wenn Sie beispielsweise das Kontrollfeld für die Einrichtung von ODBC-Verbindungen aus Access heraus anzeigen möchten, muss VBA passen. Microsoft stellt jedoch für diese Windows-Funktionen eine leistungsfähige Bibliothek "Shell Controls and Automation" zur Verfügung, die Sie auch problemlos mit Access nutzen können. Dieser zweiteilige Tipp der Woche vermittelt Grundlagen und zeigt praktische Beispiele für den Einbau in Ihre Datenbanken.

Um die Bibliothek "Shell Controls and Automation" per VBA nutzen zu können, müssen Sie zunächst einen Verweis auf das entsprechende Objektmodell setzen:

1. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung oder öffnen Sie ein Modul im Entwurfsmodus.

2. Wählen Sie das Menü Extras-Verweise an.

3. Lokalisieren Sie den Eintrag "Microsoft Shell Controls And Automation" und aktivieren Sie das Kontrollkästchen vor dem Eintrag.

4. Übernehmen Sie die Änderungen mit einem Klick auf Ok.
Falls "Microsoft Shell Controls And Automation" nicht vorhanden:

 FRAGE:  Ihren Tipp hätte ich gerne ausprobiert, doch leider wird im Bereich Verweise der Eintrag "Microsoft Shell Controls And Automation" nicht angezeigt. Was ist zu tun?
G. Brezler

 ANTWORT:  Klicken Sie auf die Schaltfläche Durchsuchen bzw. Hinzufügen, lokalisieren Sie die Datei SHELL32.DLL im Verzeichnis SYSTEM32 Ihrer Windows-Installation und übernehmen Sie sie mit einem Doppelklick. Klicken Sie dann auf die Schaltfläche Ok, um die Änderung zu übernehmen.
Um nun beispielsweise das ODBC-Kontrollfeld aufzurufen, gehen Sie folgendermaßen vor:

· Legen Sie ein neues Formular "Test" oder ähnlich.

· Platzieren Sie im Formular eine Schaltfläche ODBC-Verbindungen (Namen = btnODBC) an.

· Stellen Sie die Eigenschaft "Beim Klicken" auf den Eintrag "Ereignisprozedur]" ein und klicken Sie auf die Schaltfläche mit den drei Punkten, um den VBA-Editor zu öffnen.

· Geben Sie die folgenden Anweisungen ein:

Private Sub btnODBC_Click()
  Dim s As New Shell

  s.ControlPanelItem ("odbccp32.cpl")

  Set s = Nothing

End Sub
· Speichern Sie die Änderungen und lassen Sie das Formular anzeigen.

Nach einem Klick auf die Schaltfläche ODBC-Verbindungen wird nun der ODBC-Dialog angezeigt, so dass direkt eine ODBC-Verbindung eingerichtet oder eine vorhandene ODBC-Verbindung bearbeitet oder gelöscht werden kann.

Die Funktionsweise ist schnell erklärt: "Dim s As New Shell" initialisiert eine Objektvariable vom Datentyp "Shell", wobei "As New..." dafür sorgt, dass das Objekt direkt genutzt werden kann. Die dann folgende Anweisung "s.ControlPanelItem ("odbccp32.cpl")" ruft die Methode "ControlPanelItem" des Objektes "Shell" auf. Als Parameter ist der Dateiname des anzuzeigenden Kontrollfeldes (Dateierweiterung = CPL, Control Panel) anzugeben, der in diesem Fall "odbccp32.cpl" lautet. "Set s = Nothing" dereferenziert die Objektvariable anschließend wieder, so dass der von ihr belegte Speicherplatz wieder freigegeben wird. Ähnlich einfach können Sie auch andere Kontrollfelder wie zum Beispiel für die Einstellung von Datum und Zeit ("timedate.cpl") oder für die Regions- und Sprachoptionen ("intl.cpl") anzeigen. Eine Übersicht der verfügbaren Kontrollfelder und deren Dateinamen erhalten Sie am einfachsten über den Windows-Explorer: Lokalisieren Sie das Verzeichnis "Windows\System32" und lassen Sie dessen Inhalt in der Ansicht "Details" nach Typ sortiert anzeigen. Scrollen Sie dann nach unten, bis der Abschnitt "Systemsteuerungsoptionen" erreicht hier. Was sich hinter den einzelnen CPL-Dateien verbrigt, bringen Sie am einfachsten durch einen Doppelklick auf die Datei in Erfahrung: Daraufhin wird das Kontrollfeld mit der Systemsteuerung geöffnet. Alternativ können Sie eine CPL-Datei mit der rechten Maustaste anklicken und Mit der Systemsteuerung öffnen im Kontextmenü anwählen.

Noch ein Beispiel für nützliche Funktionen dieser Bibliothek: Eine der relativ oft gestellten Fragen in der Access-Helpline ist "Wie kann ich Access per Schaltfläche beenden und gleichzeitig den Rechner herunterfahren?". Die Antwort liegt in "Shell Controls And Automation" verborgen: Mit der Methode "ShutdownWindows" können Sie den normalerweise per Menü Start-Herunterfahren aufgerufenen Dialog anzeigen lassen. Ein Klick auf Ausschalten fährt dann den Rechner sofort herunter. Ein explizites Beenden von Access ist nicht notwendig, da Windows das beim Herunterfahren automatisch erledigt. Den Einbau einer solchen Schaltfläche nehmen Sie wie folgt vor:

· Öffnen Sie das betreffende Formular, beispielsweise ein Hauptmenü, im Entwurfsmodus.

· Platzieren Sie im Formular eine Schaltfläche Access beenden, Rechner herunterfahren (Namen = btnShutdown) an.

· Stellen Sie die Eigenschaft "Beim Klicken" auf den Eintrag "Ereignisprozedur]" ein und klicken Sie auf die Schaltfläche mit den drei Punkten, um den VBA-Editor zu öffnen.

· Geben Sie die folgenden Anweisungen ein:

Private Sub btnShutDown_Click()
  Dim s As New Shell

  s.ShutdownWindows

End Sub
· Speichern Sie die Änderungen und lassen Sie das Formular anzeigen.

Nach einem Klick auf die Schaltfläche Access beenden, Rechner herunterfahren wird nun der bekannte Dialog angezeigt. Im zweiten Teil dieses Tipps der Woche in der nächsten Ausgabe von Access Weekly stellen wir weitere nützliche Funktionen von "Shell Controls And Automation" vor. Unter anderem zeigen wir, wie Sie ein Explorer-Fenster für ein bestimmtes Verzeichnis oder eine Ordnerauswahl mit einer Schaltfläche "Neuer Ordner" anzeigen lassen.

Mit der Bibliothek "Shell Controls and Automation" stellt Windows spezielle Funktionen bereit, mit deren Hilfe Sie Ihren Datenbanken besonderen Komfort verleihen können. So hatten wir im ersten Teil dieses Tipps der Woche gezeigt, wie Sie beispielsweise den Dialog für ODBC-Verbindungen anzeigen oder Windows und Access per Mausklick einfach beenden können. Für "Späteinsteiger" noch einmal kurz der Weg, um "Shell Controls and Automation" in Access nutzen zu können:

1. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung oder öffnen Sie ein Modul im Entwurfsmodus.

2. Wählen Sie das Menü Extras-Verweise an.

3. Lokalisieren Sie den Eintrag "Microsoft Shell Controls And Automation" und aktivieren Sie das Kontrollkästchen vor dem Eintrag.

4. Übernehmen Sie die Änderungen mit einem Klick auf Ok.

5. In Prozeduren, Funktionen oder Ereignisprozeduren deklarieren Sie dann eine Objektvariable vom Datentyp "Shell" mit dem Zusatz "As New" und können so die Methoden des Objektes direkt nutzen:

Dim s As New Shell

s.ControlPanelItem "odbccp32.cpl"
Set s = Nothing

Eine Funktion, die in Datenbanken immer wieder hilfreich sein kann, ist die Anzeige eines bestimmten Verzeichnisses im Windows-Explorer. So können Sie beispielsweise dem Anwender die Möglichkeit geben, in einem Verzeichnis mit Import-Dateien die dort vorhandenen Excel-Arbeitsmappen zunächst per Doppelklick mit Excel einsehen und prüfen oder über die Funktionen des Windows-Explorers Dateien umbenennen oder Sicherheitskopien davon anlegen zu können. Um eine entsprechende Schaltfläche in ein Formular einzubauen, gehen Sie wie folgt vor:

1. Öffnen Sie das betreffende Formular im Entwurfsmodus.

2. Platzieren Sie im Formular eine Schaltfläche Import-Verzeichnis (Namen = btnImportFolder) an.

3. Stellen Sie die Eigenschaft "Beim Klicken" auf den Eintrag "[Ereignisprozedur]" ein und klicken Sie auf die Schaltfläche mit den drei Punkten, um den VBA-Editor zu öffnen.

4. Geben Sie die folgenden Anweisungen ein:

Private Sub btnImportFolder_Click()
  Dim s As New Shell

  Const cstrDir = "Z:\Test\Import"

  s.Open cstrDir

  Set s = Nothing

End Sub
· Speichern Sie die Änderungen und lassen Sie das Formular anzeigen.

Nach einem Klick auf die Schaltfläche Import-Verzeichnis wird nun ein Windows-Explorer-Fenster mit dem in der Konstanten "cstrDir" festgelegten Verzeichnis geöffnet. Der Windows-Explorer zeigt dabei ein einfaches Fenster an. Alternativ können Sie ein Windows-Explorer-Fenster mit hierarchischer Verzeichnis-Anzeige links und dem gewünschten Verzeichnis rechts anzeigen lassen. Dazu nutzen Sie die Methode "Explore" des Shell-Objekts wie in folgender Ereignisprozedur:

Private Sub btnExploreImport_Click()
  Dim s As New Shell

  Const cstrDir = "Z:\Test\Import"

  s.Explore cstrDir

  Set s = Nothing

End Sub

Neben einer Funktion zur Anzeige von Verzeichnisinhalten wird oftmals ein Dialog für die Auswahl eines Verzeichnisses oder einer Datei benötigt. Dazu hatten wir bereits Funktionen vorgestellt, die auf API-Funktionen von Windows oder auf dem mit Access 2002/XP eingeführten Objekt "FileDialogs" basieren. Alternativ können Sie dazu auch das "Shell"-Objekt nutzen, das zusätzlich in der Ordnerauswahl eine Schaltfläche "Neuer Ordner" zur Verfügung stellt. Die folgende Ereignisprozedur "Beim Klicken" einer Schaltfläche "btnChooseFolder" zeigt eine Ordnerauswahl und den gewählten Ordner per "MsgBox" an:

Private Sub btnChooseFolder_Click()
  Dim s As New Shell
  Dim f As Folder3

  Set f = s.BrowseForFolder(Me.hwnd, _
            "Bitte Import-Ordner wählen:", _
            0&, _
            "Z:\Test")

  If Not f Is Nothing Then
    MsgBox f.Self.Path
  End If

  Set f = Nothing
  Set s = Nothing

End Sub

Der erste Parameter von "BrowseForFolder" spezifiziert ein Fenster, in dessen Kontext der Auswahldialog angezeigt werden soll. Mit "Application.hWndAccessApp" legen wir hier fest, dass es sich hier um das Access-Fenster handeln soll. Alternativ können Sie auch "Me.hWnd" verwenden, um das aktuelle Formular als "Mutterfenster" festzulegen. Der zweite Parameter definiert den Text, der als Prompt im Dialog angezeigt werden soll. Der dritte Parameter übergibt spezielle Flags zur Steuerung der Anzeige des Dialogs - für die einfache Ordnerauswahl werden keine speziellen Flags benötigt. Mit dem vierten Parameter teilen Sie der Methode mit, welcher Ordner als Startordner vorgegeben werden soll. Das Ergebnis ist eine Referenz auf ein Objekt vom Datentyp "Folder3", über dessen Eigenschaft "Self.Path" Laufwerk und Pfad des gewählten Ordners ausgelesen werden kann.

Darüber hinaus stellt die Bibliothek "Shell Controls and Automation" noch einige kleinere Funktionen bereit, die sehr nützlich sein können: So lässt sich per "s.FileRun" der sonst über das Menü Start-Ausführen aktivierte Dialog "Anwendung ausführen" anzeigen, "s.FindFiles" ruft die Suchfunktion und "s.Help" die Hilfefunktion von Windows auf.

